

[ABOUT](#)[INDEX](#)[WRITE FOR US](#)

LISTEN DATA

MAKE YOUR DATA TELL A STORY

[HOME](#)[SAS](#)[R](#)[PYTHON](#)[DATA SCIENCE](#)[SQL](#)[EXCEL](#)[SPSS](#)[RESOURCES](#)[INFOGRAPHICS](#)

SEARCH...

GO

[Home](#) » [SAS](#) » Retain Statement in SASFollow us on **Facebook**

RETAIN STATEMENT IN SAS



10 Comments



SAS

This tutorial explains how to use retain statement in SAS. In SAS, it's a very easy and useful way to retain values with RETAIN statement.

Create Sample Data

The following program creates a sample data for demonstration -

```
data abcd;  
input x y;  
cards;  
1 25
```

Join us with 5000+ Subs**Subscribe to Free Updates**

```
1 28  
1 27  
2 23  
2 35  
2 34  
3 25  
3 29  
;  
run;
```

Uses of RETAIN Statement

The **RETAIN** statement simply copies retaining values by telling the SAS not to reset the variables to missing at the beginning of each iteration of the DATA step. If you would not use retain statement then SAS would return missing at the beginning of each iteration.

The retain statement keeps the value once assigned.

Generate Serial Number

Suppose you need to generate a serial number (or row index number) with data step.

```
data aaa;  
set abcd;  
retain z 0;
```

```
z = z + 1;  
run;
```

We can retain implicitly by using the +1 notation.

```
data aaa;  
set abcd;  
z + 1;  
run;
```

Output Dataset

x	y	z
1	25	1
1	28	2
1	27	3
2	23	4
2	35	5
2	34	6
3	25	7
3	29	8

Output Data Set

Cumulative Score

Suppose you need to calculate cumulative score. In financial data, we generally need to calculate cumulative score year to date.

```
data aaa;  
set abcd;  
retain z 0;  
z = z + y;  
run;
```

x	y	z
1	25	25
1	28	53
1	27	80
2	23	103
2	35	138
2	34	172
3	25	197
3	29	226

Output Data Set

Generate Serial Number by Group

Suppose you have a grouping variable say "region" and you need to generate a row index number by region.

```
proc sort data = abcd;  
by x;  
run;  
  
data aaa;  
set abcd;  
retain z;  
if first.x then z = 1;  
else z = z + 1;  
by x;  
run;
```

x	y	z
1	25	1
1	28	2
1	27	3
2	23	1
2	35	2
2	34	3
3	25	1
3	29	2

SAS : Retain Statement

Cumulative Score by Group

Suppose you need to calculate cumulative sale by product categories.

```
data aaa1;  
set aaa;  
retain z1;  
if first.x then z1 = y;  
else z1 = z1 + y;  
by x;  
run;
```

x	y	z	z1
1	25	1	25
1	28	2	53
1	27	3	80
2	23	1	23
2	35	2	58
2	34	3	92
3	25	1	25
3	29	2	54

SAS : Cumulative Score

z1 constitutes cumulative values of variable y by grouping variable x.

Number of Unique Observations

The number of unique rows by a group can easily be calculated with PROC FREQ and PROC MEANS. The following program explains how we can calculate number of observations in a categorical variable with Data Step.

```
data aaa2;  
  set abcd (drop = y);  
  retain z;  
  if first.x then z = 1;  
  else z = z + 1;  
  by x;  
  if last.x then output;  
run;
```

x	z
1	3
2	3
3	2

Unique Count

Suppose you have more than 1 grouping variable

```
data temp;  
input ID ID1 Score;  
cards;  
1 1 25  
1 1 26  
1 2 27  
1 2 29  
2 1 28  
2 1 29  
2 2 31  
;  
run;
```

```
data temp2;  
set temp;  
by ID ID1;  
if first.ID or first.ID1 then N = 1;  
else N+1;  
proc print;  
run;
```

When you have more than 1 grouping variable, we can use multiple FIRST. statements with OR operator to generate serial numbers.

SAS Tutorials : [100 Free SAS Tutorials](https://www.listendata.com/2015/10/retain-statement-in-sas.html)

Love this Post? Spread the Word



Facebook



LinkedIn



Twitter

 **Get Free Email Updates :**

Enter your email address

Submit

Please confirm your email address by clicking on the link sent to your Email

Related Posts:

[Check number of observations in SAS dataset](#)

[Predictive Modeling Interview Questions and Answers](#)

[Extract last 4 characters / digits of value in SAS](#)

[SAS : Calculate AUC of Validation Data](#)

[Detecting Interaction in Regression Model](#)

10 Responses to "Retain Statement in SAS"



Unknown 3 November 2015 at 05:39

Hi,
Thank you for making it so easy to understand took me just 1 min to understand the coding...great work brother.

Also can you please help me how can i prepare better for my base sas exam as it due in dec.

Reply

▼ [Replies](#)



Deepanshu Bhalla 15 January 2016 at 12:34

Author

Check out this link - [Base SAS Certification Questions and Answers](#)



boyina vikash 2 December 2017 at 20:13

Can you explain...how to find out the middlest observation in a data set?????
Please....

Reply



snlakshmi vajja 30 December 2016 at 01:03

Hi,
thank you for your clear information.
please can you give me the answer for this question.
how can we find out cumulative totals for salary on each by group of gender?

Reply

▼ Replies



Deepanshu Bhalla 30 December 2016 at 02:36

Author

Check the code below -

```
data abcd;  
input Gender$ Salary;  
cards;  
M 25  
M 26  
M 27  
F 23  
F 24  
F 25  
;  
run;
```

```
proc sort data = abcd;  
by gender;  
run;
```

```
data temp;  
set abcd;  
by gender;  
retain tot ;
```

```
if first.gender then tot = salary;  
else tot = tot + salary;  
if last.gender;  
drop salary;  
run;
```

Hope it helps!



Rp Vadla 8 May 2018 at 21:30

```
data ds;  
set sashelp.class;  
run;  
proc sort data=ds;  
by sex;  
run;  
data ds;  
set ds;  
by sex;  
retain cum_height 0 cum_weight 0;  
if first.sex then cum_height=height;  
else cum_height=cum_height+height;  
if first.sex then cum_weight=weight;  
else cum_weight=cum_weight+weight;  
run;
```

i think this may help you

Reply



Sivanagalakshmi .vajja 30 December 2016 at 03:03

This is not what i want.

Reply

▼ Replies



Deepanshu Bhalla 30 December 2016 at 03:07

Author

Please post your sample data with desired output. It's hard to guess the requirement. Thanks!

[Reply](#)**Unknown** 24 May 2017 at 23:19

Hi Deepanshu,

Could you please share Advance sas certification questions and answers as well..

Thanks!!

[Reply](#)**Unknown** 14 August 2018 at 11:03

Can you tell me
how to create data lines in missing values?

[Reply](#)

Enter your comment...



Comment as: Google Account

[Publish](#)[Preview](#)[← PREV](#)[NEXT →](#)

Copyright 2018 [ListenData](#)