

# Great Graphics Using Proc Sgplot, Proc Sgscatter, and ODS Graphics for SAS®/Stat Procedures

Kathy Welch

CSCAR

The University of Michigan

MSUG Meeting, Tuesday April 27, 2010

# What we will Cover

- Introduction to Statistical Graphics Procedures
  - Proc Sgplot
  - Proc Sgscatter
  - Proc Sgpanel
- Editing ODS graphics files
- Examples of ODS graphics with Statistical Procedures
  - Proc ttest
  - Proc Reg
  - Proc GLM
  - Proc Logistic


# Statistical Graphics Using Proc Sgplot, Proc Sgscatter and Proc Sgpanel

- Statistical graphics plots use ODS (output delivery system) graphics
- Statistical graphics are easy to produce, look nice, and are more intuitive than traditional SAS/Graph graphics
- Statistical Graphics can be edited (to some extent) interactively

# Where are my graphs?

- Graphs created with ODS graphics will be in Results Window, not in Graph Window
- Double-click on the graphics icon to view the file, using local windows graphics viewer.
- Graphs will be automatically be saved as .png files in current SAS folder
- png (portable network graphics) files are
  - Raster graphics
  - Compact format
  - Usable in most windows applications

# Getting Started




2. Double-click on graphics file icon to view graph

1. Set the current folder


# Boxplots

```
proc sgplot data=mylib.employee;  
  vbox salary; run;
```




# Boxplots for Categories

```
proc sgplot data=mylib.employee;  
  vbox salary/ category=gender; run;
```



# Paneled Boxplots


```
proc sgpanel data=mylib.employeee;  
  panelby jobcat / rows=1 columns=3 ;  
  vbox salary / category= gender; run;
```






# Barcharts

```
proc sgplot data=mylib.employeee;  
  vbar jobcat ; run;
```




# Stacked Barcharts

```
proc sgplot data=mylib.employee;  
  vbar jobcat /group=Gender; run;
```




# Barcharts with Means and Error Bars

```
proc sgplot data=mylib.employeee;  
  vbar jobcat / response=salary limitstat = stddev  
  limits = upper stat=mean; run;
```




# Barcharts of Proportions

```
proc sgplot data=afifi;  
  vbar shoktype / response=died stat=mean;  
  format shoktype shokfmt.; run;
```




# Paneled Barcharts

```
proc sgpanel data=mylib.employee;  
  panelby gender ;  
  vbar jobcat / response=salary limitstat = stddev  
  limits = upper stat=mean; run;
```




# Histograms

```
proc sgplot data=mylib.employee;  
  histogram salary ;  
  density salary;  
  density salary / type=kernel;  
  keylegend / location=inside position=topright;
```



# Overlaid Histograms


```
proc sgplot data=mylib.employee;  
  histogram salbegin ;  
  histogram salary / transparency = .5; run;
```



Note: Transparency = 0 is opaque. Transparency = 1.0 is fully transparent.

# Scatterplots


```
proc sgplot data=mylib.employee;  
  scatter x=salbegin y=salary /  
    group=gender; run;
```






# Scatterplot with Regression Line

```
proc sgplot data=mylib.employee;  
  where jobcat=1;  
  scatter x=prevexp y=salary / group=gender ;  
  reg x=prevexp y=salary / cli clm nomarkers; run;
```




# Regression Lines for Subgroups

```
proc sgplot data=mylib.employee;  
  where jobcat=1;  
  reg x=prevexp y=salary / group=gender; run;
```




# Paneled Scatterplot with Loess Fit

```
proc sgpanel data=mylib.employeee;  
  panelby jobcat / columns=3;  
  scatter x=jobtime y=salary / group=gender;  
  loess x=jobtime y=salary nomarkers; run;
```




# Scatterplot Matrix

```
proc sgscatter data=mylib.employeee;  
  where jobcat=1;  
  matrix salbegin salary jobtime prevexp / group=gender  
        diagonal=(histogram kernel); run;
```



# Series Plots

```
proc sgpanel data=autism;  
  panelby sicdegp /columns=3;  
  series x=age y=vsae / group=Childid  
  markers legendlabel=" " lineattrs=(pattern=1  
  color=black).
```



# Generate Means


```
proc sort data=autism;
  by sicdegp age;
run;

proc means data=autism noprint;
  by sicdegp age;
  output out=meandat mean(VSAE)=mean_VSAE;
run;

data autism2;
  merge autism meandat(drop=_type__freq_);
  by sicdegp age;
run;
```

# Mean Plots Overlaid on Raw Data

```
proc sgplot data=autism2;  
  series x=age y=mean_VSAE / group=SICDEGP;  
  scatter x=age y=VSAE ; run;
```



# Formats Make Graphs More Readable


```
proc format;  
  value jobcat 1="Clerical"  
              2="Custodial"  
              3="Manager";  
  value $Gender "f"="Female"  
              "m"="Male";  
run;
```



# Formats Make Graphs More

## Readable2

```
proc sgpanel data=mylib.employeee;  
  panelby jobcat / rows=1 columns=3 novarname;  
  vbox salary / category= gender ;  
  format gender $gender. jobcat jobcat; run;
```




# Editing ODS Graphics

- Use “sgedit”, or “sgedit on” to enable ODS graphics editor
  - Submit in command dialog box
- Two outputs will be created for each graph.
  - .png file
  - .sge (SAS Graphics Editor) file
- Edit the .sge file, and save as a .png to use in other applications.

# Some of the Things you can Edit with the ODS Graphics Editor

- Title, footnote
- Axis labels, colors, fonts, symbols, text boxes
- Graph Style (Try them)
  - Listing (the default, colors)
  - Analysis (color scheme)
  - Journal (grayscale)
  - Journal2 (black and white)
  - Statistical (color scheme)

# SAS Graphics Editor Window




# Creating PDF Output

- Send output to a pdf file
- Similar for HTML or rtf file output

```
ods pdf style=journal2;  
ods pdf file = "testing.pdf";  
ods listing close;  
title "PDF Output";  
proc sgpanel data=mylib.employee;  
    panelby jobcat;  
    scatter x=jobtime y=salary / group=gender;  
    loess x=jobtime y=salary / nomarkers ; run;  
ods pdf close;  
ods listing;
```

PDF Output will be in the Results Viewer Window



# Where to Get Help for Statistical Graphics Procedures

- SAS Help and Documentation > Contents > SAS Products > SAS/Graph > SAS/Graph 9.2 Statistical Procedures Guide > SAS/Graph Statistical Graphics Procedures
- Once you find it, it's really helpful.


# ODS Graphics in SAS/Stat Procedures

- ODS graphics are available for statistical procedures
- Submit “ods graphics on;” before the procedure
- Submit “ods graphics off;” after the procedure
- Graphs are .png files, as for Proc Sgplot, Sgscatter and Sgpanel.
- ODS graphics show up after “drilling down” in the Results window for the procedure.
- Check SAS documentation for available ODS graphics for each procedure





# Proc ttest

```
ods graphics on;  
proc ttest data=mylib.afifi;  
class survive;  
var sbp1;  
run;  
ods graphics off;
```





# Proc Reg

```
ods graphics on;
title "Regression with Default Plots";
proc reg data=mylib.afifi;
model sbp2 = sbp1 sex age/partial;
run; quit;
ods graphics off;
```



# Proc Reg Unpack Graphics


proc reg plots(only)=DiagnosticsPanel(unpack);



# Proc GLM

```
ods graphics on;  
title "Anova with Plots";  
proc glm data=mylib.afifi;  
  class shoktype ;  
  model sbp1 = shoktype;  
  lsmeans shoktype / pdiff adjust=tukey;  
run; quit;  
ods graphics off;
```



# Proc GLM DiffPlot



# Proc Logistic

```
ods graphics on;  
proc logistic data = mylib.afifi descending  
plots(only)=(effect roccurve) ;  
class shoktype;  
model survive = sbp1 shoktype;  
output out=predadat p=predict reschi=rchi  
resdev=rdev  
      h=leverage;  
run; quit;  
ods graphics off;
```

# Proc Logistic ROC Curve and Effect Plot



# Thank you!

- Try some of these graphs using your data.
- They're easy and fun!